

Конструктор запросов в управляемом приложении в «1С:Предприятие 8.3.5»

В платформе 8.3.5.823 появилась давно ожидаемая разработчиками возможность – был реализован конструктор запроса, работающий в режиме управляемого приложения.

Теперь конструктор запроса доступен в тонком и толстом клиентах, а также в веб-клиенте.

Знакомство с новым механизмом начнем с написания простой обработки, которая будет открывать конструктор в тонком клиенте.

Создадим внешнюю обработку, на форме разместим поле ввода для текста запроса и кнопку, открывающую конструктор:



Напишем следующий обработчик нажатия на кнопку:

&НаКлиенте

Процедура Конструктор (Команда)

```
Конструктор = Новый КонструкторЗапроса;  
Конструктор.Текст = ТекстЗапроса;
```

```
Оповещение = Новый ОписаниеОповещения ("ОткрытьКонструкторЗапросаЗавершение",  
ЭтотОбъект);
```

```
Конструктор.Показать (Оповещение);
```

КонецПроцедуры

&НаКлиенте

Процедура ОткрытьКонструкторЗапросаЗавершение (Текст, ДополнительныеПараметры) Экспорт

Если НЕ Текст = Неопределено Тогда


ТекстЗапроса = Текст;

КонецЕсли;

КонецПроцедуры


Обратите внимание, что вместо метода *ОткрытьМодально()* использовался новый метод *Показать()*, который открывает конструктор запроса немодально.

При открытии обработки в тонком клиенте и нажатии на кнопку открывается форма конструктора запроса:




По сравнению с классическим конструктором, к которому мы привыкли еще со времен 8.0, внешний вид нового претерпел несколько изменений.

Первое, что бросается в глаза - отсутствие привычных закладок, отображающих запросы пакета, имена временных таблиц и объединения запросов:




Вместо этого теперь в нижней части окна расположены два выпадающих списка – “Редактируемый запрос” и “Редактируемый оператор”.




В первом из них перечислены запросы, входящие в пакет, для временных таблиц указываются имена. Во втором списке можно выбрать, который из объединяемых запросов редактируется в данный момент.

В классическом конструкторе запроса для создания запроса на уничтожение временной таблицы нужно было установить переключатель на закладке “Дополнительно” в соответствующую позицию:




А при использовании управляемого конструктора запроса это действие выполняется на закладке “Пакет запросов” при выборе соответствующего пункта кнопки “Добавить”:



Но все-таки самые большие отличия расположены на закладке “Связи”. Теперь чтобы связать две таблицы, необходимо мышкой перетащить одну таблицу поверх другой.

Таблицы после этого будут отображаться в виде дерева. Далее необходимо указать вид соединения в выпадающем списке (внутреннее, левое, правое, полное) и написать условие соединения.



Вид соединения теперь выбирается явно из списка, а не указывается галочками.

Также больше не существует галочки “Произвольное условие” при описании связей и условий, все условия редактируются только в редакторе произвольного выражения или вручную.

Схема запроса

На практике часто встречается необходимость сформировать различный текст запроса в зависимости от настроек параметров учета в конфигурации.

Для этого зачастую применяется следующий подход: строка, содержащая текст запроса, собирается динамически, “по кусочкам”.

Для примера рассмотрим, как в типовой конфигурации “Управление торговлей, редакция 10.3” формируется текст запроса.

```
ТекстЗапроса = "  
|ВЫБРАТЬ // Запрос, контролирующий остатки на складах  
| Док.Номенклатура.Представление КАК НоменклатураПредставление,  
...  
| %ПОЛЕ_Док_Серия% КАК СерияНоменклатуры,  
...  
| ЕСТЬNULL (МАКСИМУМ (Резервы.КоличествоОстаток) , 0) КАК РезервыКоличество  
| ИЗ  
| "+ ТекстЗапросаРеквизитыДокумента + "  
|  
|ЛЕВОЕ СОЕДИНЕНИЕ  
| РегистрНакопления.ТоварыВРезервеНаСкладах.Остатки (,Склад в (&СписокСкладов) И  
| ДокументРезерва в (&СписокДокументовРезерва) И  
| %ВыборкаПоНоменклатуре%) КАК Резервы  
| ПО  
| Док.Номенклатура = Резервы.Номенклатура  
| И Док.Склад = Резервы.Склад  
| И Док.ЗаказПокупателя = Резервы.ДокументРезерва  
| //СОЕДИНЕНИЕ_Характеристика_Резервы  
| //СОЕДИНЕНИЕ_Серия_Резервы  
...  
| ";  
  
ТекстЗапроса = СтрЗаменить (ТекстЗапроса, "%ВыборкаПоНоменклатуре%", "Номенклатура В  
("+ТекстЗапросаСписокНоменклатуры+")");  
...  
Если не УчитыватьСерии Тогда  
ТекстЗапроса = СтрЗаменить (ТекстЗапроса, "%ПОЛЕ_Док_Серия%", "&ПустаяСерия");  
КонецЕсли;
```

Как видно, текст запроса собирается из частей при помощи конкатенации, в тексте запроса намеренно используются недопустимые символы (например, %).

Части текста запроса подменяются при помощи *СтрЗаменить()* на те или иные значения в зависимости от различных условий.

При этом воспользоваться конструктором запроса невозможно – такой текст он не распознает, редактировать запрос в таком случае придется вручную, а программный код получается достаточно громоздким.

В платформе 8.3.5 появилась объектная модель схемы запроса. Теперь программно можно формировать текст запроса, пользуясь не конкатенацией отдельных строк, а оперируя объектами встроенного языка.

Для этого был реализован специальный объект *СхемаЗапроса*. Для примера составим при помощи объектной модели на демо-конфигурации “Управляемое приложение” текст запроса, в котором к табличной части документа “Расход товара” при помощи левого соединения присоединяем срез последних регистра сведений “Цены товаров”.

Пояснения к разрабатываемому коду даны в комментариях.

```
СхемаЗапроса = Новый СхемаЗапроса;
//при создании схема содержит один пакет и один оператор в пакете.
Пакет = СхемаЗапроса.ПакетЗапросов [0];
Оператор = Пакет.Операторы [0];

//получаем таблицы для дальнейшего использования
ДоступнаяТаблицаРасходТовара =
Пакет.ДоступныеТаблицы.Найти ("Документ.РасходТовара");
ДоступнаяТаблицаРасходТовараТовары =
Пакет.ДоступныеТаблицы.Найти ("Документ.РасходТовара.Товары");

//добавляем источники
Источник = Оператор.Источники.Добавить (ДоступнаяТаблицаРасходТовараТовары);
Источник.Источник.Псевдоним = "РасходТовараТовары";

Источник =
Оператор.Источники.Добавить (Пакет.ДоступныеТаблицы.Найти ("РегистрСведений.ЦеныТоваров.
СрезПоследних"));
Источник.Источник.Псевдоним = "ЦеныТоваровСрезПоследних";

//задаем параметры виртуальной таблицы для среза последних регистра сведений
Источник.Источник.Параметры [0].Выражение = Новый ВыражениеСхемыЗапроса ("&Дата");
Источник.Источник.Параметры [1].Выражение = Новый ВыражениеСхемыЗапроса ("ВидЦен =
&ВидЦен");

//устанавливаем условия левого соединения таблиц
Соединение = Оператор.Источники [0].Соединения.Добавить ("ЦеныТоваровСрезПоследних",
"РасходТовараТовары.Товар = ЦеныТоваровСрезПоследних.Товар");
Оператор.Источники [0].Соединения [0].ТипСоединения =
ТипСоединенияСхемыЗапроса.ЛевоеВнешнее;

//перечисляем выходные поля
//для первых двух псевдоним будет сгенерирован автоматически
//а для третьего принудительно укажем псевдоним "Цена"
ВыбранноеПоле = Оператор.ВыбираемыеПоля.Добавить ("РасходТовараТовары.Товар");
ВыбранноеПоле = Оператор.ВыбираемыеПоля.Добавить ("РасходТовараТовары.Количество");
ВыбранноеПоле =
Оператор.ВыбираемыеПоля.Добавить ("ЕСТЬNULL (ЦеныТоваровСрезПоследних.Цена, 0)");

Пакет.Колонки [2].Псевдоним = "Цена";

//добавляем секцию ГДЕ в запрос
Оператор.Отбор.Добавить ("РасходТовараТовары.Ссылка = &Ссылка");

Сообщить (СхемаЗапроса.ПолучитьТекстЗапроса ());
```

При выполнении метода *ПолучитьТекстЗапроса()* объекта *СхемаЗапроса* мы получим следующий текст запроса:

ВЫБРАТЬ

```
РасходТовараТовары.Товар,  
РасходТовараТовары.Количество,  
ЕСТЬNULL (ЦеныТоваровСрезПоследних.Цена, 0) КАК Цена
```

ИЗ

```
Документ.РасходТовара.Товары КАК РасходТовараТовары  
ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЦеныТоваров.СрезПоследних (&Дата, ВидЦен =  
&ВидЦен) КАК ЦеныТоваровСрезПоследних  
ПО (РасходТовараТовары.Товар = ЦеныТоваровСрезПоследних.Товар)
```

ГДЕ

```
РасходТовараТовары.Ссылка = &Ссылка
```

Если же необходимо готовый текст запроса трансформировать в объект *СхемаЗапроса*, то следует воспользоваться методом *УстановитьТекстЗапроса()*:

```
Текст = "ВЫБРАТЬ
```

```
| РасходТовараТовары.Товар,  
| РасходТовараТовары.Количество,  
| ЕСТЬNULL (ЦеныТоваровСрезПоследних.Цена, 0) КАК Цена  
| ИЗ  
| Документ.РасходТовара.Товары КАК РасходТовараТовары  
| ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЦеныТоваров.СрезПоследних (&Дата,  
ВидЦен = &ВидЦен) КАК ЦеныТоваровСрезПоследних  
| ПО РасходТовараТовары.Товар = ЦеныТоваровСрезПоследних.Товар  
| ГДЕ  
| РасходТовараТовары.Ссылка = &Ссылка";
```

```
СхемаЗапроса = Новый СхемаЗапроса;  
СхемаЗапроса.УстановитьТекстЗапроса (Текст);
```

После этого объект *СхемаЗапроса* будет содержать структуру запроса в виде набора свойств.

Использование схемы запроса позволяет редактировать текст запроса более наглядно, оперируя при этом элементами структуры запроса.

В таком случае возможно избежать получения запроса при помощи сборки из отдельных фрагментов, что должно повысить наглядность и читаемость кода.

Будем ждать, когда разработчики типовых решений начнут применять объектную модель запросов при написании конфигураций. Это позволит нам на практике познакомиться с применением новых возможностей платформы 8.3.5.

Ханевич Василий

г. Калининград

Дополнительные материалы

Все статьи проекта **Курсы-по-1С.рф**: <http://курсы-по-1с.рф/blog/articles/>

Курсы по программированию в 1С v.8

Базовый и Продвинутой курсы по Программированию на Платформе 1С 8
<http://www.Spec8.ru/>



Базовый курс по программированию в 1С v.8

Курс про **готовые приемы и решения**
90% задач по программированию в 1С



Продвинутой курс по программированию в 1С v.8

Больше, чем Вы можете себе представить
Детальнее требований на **1С:Специалист**

«Курс по подготовке к Аттестации по Платформе 1С 8.2 / 8.3»

<http://курсы-по-1с.рф/dev-attestation/>



Подготовка к Аттестации по Платформе 1С v.8

Аттестация по Платформе – **с первого раза**
Экономия 100 - 150 часов подготовки